

## Section 10

# System tasks and functions

--Martin O'Leary, Cadence Design Systems, 6/9/2003

*NOTE: changes are all in section 10.9 which is a totally new section*

This section describes system tasks and functions available in Verilog-AMS HDL.

### 10.1 Environment parameter functions

The syntax for these functions are shown in Syntax 10-1.

```
environment_parameter_functions ::=
    $temperature
    | $abstime
    | $realtime [ ( real_number ) ]
    | $vt [ ( temperature_expression ) ]
```

Syntax 10-1—Syntax for the environment parameter functions

These functions return information about the current environment parameters as a real value.

**\$temperature** does not take any input arguments and returns the circuit's ambient temperature in Kelvin units.

**\$abstime** returns the absolute time, that is a real value number representing time in seconds.

**\$realtime** can have an optional argument which scales the time. If no argument is given, **\$realtime**'s return value is scaled to the ``time_unit` of the module which invoked it. If an argument is given, **\$realtime** shall divide the absolute time by the value of the argument (i.e., scale to the value specified in the argument). The argument for **\$realtime** follows the semantics of the ``time_unit`, that is it shall consist of an integer followed by a scale factor. Valid integers are: 1, 10, and 100; valid scale factors are: `s` (seconds), `ms` (milliseconds), `us` (microseconds), `ns` (nanoseconds), `ps` (picoseconds), and `fs` (femtoseconds).

**\$vt** can optionally have temperature (in Kelvin units) as an input argument and returns the thermal voltage ( $kT/q$ ) at the given temperature. **\$vt** without the optional input temperature argument returns the thermal voltage using **\$temperature**.

**Note:** Previous Verilog-A modules using `$realttime` may not produce the same results as in the past as `$realttime` used to return time scaled to one (1) second. If the `time_unit` of the ``timescale` directive is set to `1s`, the behavior shall be the same. For analog blocks, the `$abstime` function should typically be used, as it returns time in seconds.

## 10.2 \$random function

The syntax for this function is shown in Syntax 10-2.

```
random_function ::=
    $random [ ( seed_expression ) ] ;
```

Syntax 10-2—Syntax for the `random_function`

`$random` provides a mechanism for generating random numbers. The function returns a new 32-bit random number each time it is called. The random number is a signed integer; it can be positive or negative.

`seed_expression` controls the numbers `$random` returns. `seed` shall be a **reg**, **integer**, or **time** variable. The `seed` value shall be assigned to this variable prior to calling `$random`.

*Examples:*

Where  $b > 0$ , the expression `( $random % b )` gives a number in the following range: `[(-b+1) : (b-1)]`.

The following code fragment shows an example of random number generation between -59 and 59:

```
integer rand;
rand = $random % 60;
```

## 10.3 \$dist\_ functions

The syntax for these functions are shown in Syntax 10-3.

```

distribution_functions ::=
    $digital_dist_functions ( args ) ;
| $rdist_uniform ( seed, start_expression, end_expression ) ;
| $rdist_normal ( seed, mean_expression, standard_deviation_expression ) ;
| $rdist_exponential ( seed, mean_expression ) ;
| $rdist_poisson ( seed, mean_expression ) ;
| $rdist_chi_square ( seed, degree_of_freedom_expression ) ;
| $rdist_t ( seed, degree_of_freedom_expression ) ;
| $rdist_erlang ( seed, k_stage_expression, mean_expression ) ;

seed ::=
    integer_variable_identifier

```

Syntax 10-3—Syntax for the probabilistic distribution functions

The following rules apply to these functions.

- All parameters to the system functions are real values, except for *seed* (which is defined by \$random()). For the \$rdist\_exponential, \$rdist\_poisson, \$rdist\_chi\_square, \$rdist\_t, and \$rdist\_erlang functions, the parameters *mean*, *degree\_of\_freedom*, and *k\_stage* shall be greater than zero (0).
- Each of these functions returns a pseudo-random number whose characteristics are described by the function name, e.g., \$rdist\_uniform returns random numbers uniformly distributed in the interval specified by its parameters.
- For each system function, the *seed* parameter is an inout parameter; that is, a value is passed to the function and a different value is returned. The system functions shall always return the same value given the same *seed*. This facilitates debugging by making the operation of the system repeatable. The argument for *seed* shall be an integer variable, which is initialized by the user and only updated by the system function. This ensures the desired distribution is achieved.
- All functions return a real value.
- In \$rdist\_uniform, the *start* and *end* parameters are real inputs which bound the values returned. The *start* value shall be smaller than the *end* value.
- The *mean* parameter used by \$rdist\_normal, \$rdist\_exponential, \$rdist\_poisson, and \$rdist\_erlang is an real input which causes the average value returned by the function to approach the value specified.
- The *standard\_deviation* parameter used by \$rdist\_normal is a real input, which helps determine the shape of the density function. Using larger numbers for *standard\_deviation* spreads the returned values over a wider range. Using a *mean* of zero (0) and a *standard\_deviation* of one (1), \$rdist\_normal generates Gaussian distribution.

- The *degree\_of\_freedom* parameter used by `$rdist_chi_square` and `$rdist_t` is a real input, which helps determine the shape of the density function. Using larger numbers for *degree\_of\_freedom* spreads the returned values over a wider range.

## 10.4 Simulation control system tasks

There are two simulation control system tasks, `$finish` and `$stop`.

### 10.4.1 `$finish`

The syntax for this task is shown in Syntax 10-4.

```
finish_task ::=
  $finish [ ( n ) ] ;
```

Syntax 10-4—Syntax for the *finish\_task*

`$finish` simply makes the simulator exit. If an expression is supplied to this task, its value determines which diagnostic messages are printed before the prompt is issued, as shown in Table 10-1. One (1) is the default if no argument is supplied.

Table 10-1—Diagnostic messages

Parameter	Message
0	Prints nothing
1	Prints simulation time and location
2	Prints simulation time, location, and statistics about the memory and CPU time used in simulation

### 10.4.2 `$stop`

The syntax for this task is shown in Syntax 10-5.

```
stop_task ::=
  $stop [ ( n ) ] ;
```

Syntax 10-5—Syntax for the *stop\_task*

`$stop` causes simulation to be suspended at a converged timepoint. This task takes an optional expression argument (0, 1, or 2), which determines what type of diagnostic message is printed. The amount of diagnostic messages output increases with the value of *n*, as shown in Table 10-1.

## 10.5 File operation tasks

This section details the file operation tasks.

### 10.5.1 \$fopen

The syntax for this task is shown in Syntax 10-6.

```
file_open_task ::=
    integer multi_channel_descriptor = $fopen ( " file_name " );
```

Syntax 10-6—Syntax for the *file\_open\_task*

**\$fopen** opens the file specified as an argument and returns a 32-bit multichannel descriptor which is uniquely associated with the file. It returns 0 if the file could not be opened for writing.

The multichannel descriptor can be thought of as a set of 32 flags, where each flag represents a single output channel. The least significant bit (`bit 0`) of a multichannel descriptor always refers to the standard output. The standard output is also called `channel 0`. The other bits refer to channels which have been opened by **\$fopen**.

The first call to **\$fopen** opens `channel 1` and returns a multichannel descriptor value of 2—that is, `bit 1` of the descriptor is set. A second call to **\$fopen** opens `channel 2` and returns a value of 4—that is, only `bit 2` of the descriptor is set. Subsequent calls to **\$fopen** open `channels 3, 4, 5`, and so on and return values of 8, 16, 32, and so on, up to a maximum of 32 open channels. Thus, a channel number corresponds to an individual bit in a multichannel descriptor.

### 10.5.2 \$fclose

The syntax for this task is shown in Syntax 10-7.

```
file_close_task ::=
    $fclose ( multi_channel_descriptor_identifier );
```

Syntax 10-7—Syntax for the *file\_close\_task*

**\$fclose** closes the channels specified in the multichannel descriptor and does not allow any further output to the closed channels. **\$fopen** reuses channels which have been closed.

## 10.6 Display tasks

The syntax for these functions are shown in Syntax 10-8.

```

display_tasks ::=
  $strobe ( list_of_arguments ) ;
| $display ( list_of_arguments ) ;
| $monitor ( list_of_arguments ) ;
| $write ( list_of_arguments ) ;

```

Syntax 10-8—Syntax for the `display_tasks`

The following rules apply to these functions.

- **\$strobe** provides the ability to display simulation data when the simulator has converged on a solution for all nodes.
- **\$strobe** displays its arguments in the same order they appear in the argument list. Each argument can be a quoted string, an expression which returns a value, or a null argument.
- The contents of string arguments are output literally, except when certain escape sequences are inserted to display special characters or specify the display format for a subsequent expression.
- Escape sequences are inserted into a string in three ways:
  - The special character `\` indicates the character to follow is a literal or non-printable character (see Table 10-2).
  - The special character `%` indicates the next character shall be interpreted as a format specification which establishes the display format for a subsequent expression argument (see Table 10-3). For each `%` character which appears in a string, a corresponding expression argument shall be supplied after the string.
  - The special character string `%%` indicates the display of the percent sign character (`%`) (see Table 10-2).
- Any null argument produces a single space character in the display. (A *null argument* is characterized by two adjacent commas (`, ,`) in the argument list.)
- When **\$strobe** is invoked without arguments, it simply prints a newline character.

The **\$display** task provides the same capabilities as **\$strobe**. The **\$write** task provides the same capabilities as **\$strobe**, but with no newline. The **\$monitor** task provides the same capabilities as **\$strobe**, but outputs only when a parameter changes.

### 10.6.1 Escape sequences for special characters

The escape sequences shown in Table 10-2, when included in a string argument, print special characters.

**Table 10-2— Escape sequences for printing special characters**

<code>\n</code>	The newline character
<code>\t</code>	The tab character
<code>\\</code>	The <code>\</code> character
<code>\"</code>	The <code>"</code> character
<code>\ddd</code>	A character specified by 1 to 3 octal digits
<code>%%</code>	The <code>%</code> character

### 10.6.2 Format specifications

Table 10-3 shows the escape sequences used for format specifications. Each escape sequence, when included in a string argument, specifies the display format for a subsequent expression. For each `%` character (except `%m` and `%%`) which appears in a string, a corresponding expression shall follow the string in the argument list, except a null argument. The value of the expression replaces the format specification when the string is displayed.

**Table 10-3— Escape sequences for format specifications**

<code>%h</code> or <code>%H</code>	Display in hexadecimal format
<code>%d</code> or <code>%D</code>	Display in decimal format
<code>%o</code> or <code>%O</code>	Display in octal format
<code>%b</code> or <code>%B</code>	Display in binary format
<code>%c</code> or <code>%C</code>	Display in ASCII character format
<code>%m</code> or <code>%M</code>	Display hierarchical name
<code>%s</code> or <code>%S</code>	Display as a string

Any expression argument which has no corresponding format specification is displayed using the default decimal format in `$stroke`.

The format specifications in Table 10-4 are used for real numbers and have the full formatting capabilities available in the C language. For example, the format specification `%10.3g` sets a minimum field width of 10 with three (3) fractional digits.

**Table 10-4— Format specifications for real numbers**

<code>%e</code> or <code>%E</code>	Display 'real' in an exponential format
<code>%f</code> or <code>%F</code>	Display 'real' in a decimal format
<code>%g</code> or <code>%G</code>	Display 'real' in exponential or decimal format, whichever format results in the shorter printed output

### 10.6.3 Hierarchical name format

The `%m` format specifier does not accept an argument. Instead, it causes the display task to print the hierarchical name of the module, task, function, or named block which invokes the system task containing the format specifier. This is useful when there are many instances of the module which call the system task. One obvious application is timing check messages in a flip-flop or latch module; the `%m` format specifier pinpoints the module instance responsible for generating the timing check message.

### 10.6.4 String format

The `%s` format specifier is used to print ASCII codes as characters. For each `%s` specification which appears in a string, a corresponding parameter shall follow the string in the argument list. The associated argument is interpreted as a sequence of 8-bit hexadecimal ASCII codes, with each 8 bits representing a single character. If the argument is a variable, its value shall be right-justified so the right-most bit of the value is the least-significant bit of the last character in the string. No termination character or value is required at the end of a string and leading zeros (0) are never printed.

## 10.7 Announcing discontinuity

The `$discontinuity` function is used to give hints to the simulator about the behavior of the module so the simulator can control its simulation algorithms to get accurate results in exceptional situations. This function does not directly specify the behavior of the module. `$discontinuity` shall be executed whenever the analog behavior changes discontinuously.

The general form is

```
$discontinuity[ ( constant_expression ) ];
```

where *constant\_expression* indicates the degree of the discontinuity.

`$discontinuity(i)` implies a discontinuity in the *i*'th derivative of the constitutive equation with respect to either a signal value or time where *i* must be non-negative.

Hence, `$discontinuity(0)` indicates a discontinuity in the equation,

`$discontinuity(1)` indicates a discontinuity in its slope, etc.

**Note:** Because discontinuous behavior can cause convergence problems, discontinuity shall be avoided whenever possible.

The filter functions (`transition()`, `slew()`, `laplace()`, etc.) can be used to smooth discontinuous behavior. However, in some cases it is not possible to implement the desired functionality using these filters. In those cases, the `$discontinuity` function shall be executed when the signal behavior changes abruptly.

**Note:** Discontinuity created by switch branches and built-in system functions, such as `transition()` and `slew()`, does not need to be announced.



*Examples:*

*Example 1*

The following example uses the discontinuity function to model a relay.

```

module relay (c1, c2, pin, nin) ;
  inout c1, c2 ;
  input pin, nin ;
  electrical c1, c2, pin, nin ;
  parameter real r=1 ;

  analog begin
    @( cross(V(pin,nin))) $discontinuity ;
    if (V(pin,nin) >= 0)
      I(c1,c2) <+ V(c1,c2)/r;
    else
      I(c1,c2) <+ 0 ;
    end
  endmodule

```

In this example, `cross()` controls the time step so the time when the relay changes position is accurately resolved. It also triggers the `$discontinuity` function, which causes the simulator to react properly to the discontinuity. This would have been handled automatically if the type of the branch `(c1,c2)` had been switched between voltage and current.

*Example 2*

Another example is a source which generates a triangular wave. In this case, neither the model nor the waveforms generated by the model are discontinuous. Rather, the waveform generated is piecewise linear with discontinuous slope. If the simulator is aware of the abrupt change in slope, it can adapt to eliminate problems resulting from the discontinuous slope (typically changing to a first order integration method).

```

module triangle(out);
  output out;
  voltage out;
  parameter real period = 10.0, amplitude = 1.0;
  integer slope;
  real offset;

  analog begin
    @( timer(0, period)) begin
      slope = +1;
      offset = $abstime ;
      $discontinuity;
    end

    @( timer(period/2, period)) begin
      slope = -1 ;
      offset = $abstime;
      $discontinuity ;
    end
  end

```

```

        V(out) <+ amplitude*slope*
            (4*($abstime - offset)/period - 1);
    end
endmodule

```

## 10.8 Time related functions

The `$bound_step()` function puts a bound on the next time step. It does not specify exactly what the next time step is, but it bounds how far the next time point can be from the present time point. The function takes the maximum time step as an argument. It does not return a value.

The general form is

```
$bound_step ( expression );
```

where *expression* is a required argument and represents the maximum timestep the simulator can advance.

*Examples:*

The example below implements a sinusoidal voltage source and uses the `$bound_step()` function to assure the simulator faithfully follows the output signal (it is forcing 20 points per cycle).

```

module vsine(out);
    output out;
    voltage out;
    parameter real freq=1.0, ampl=1.0, offset=0.0;

    analog begin
        V(out) <+ ampl*sin(2.0*`M_PI*freq*$abstime) + offset;
        $bound_step(0.05/freq);
    end
endmodule

```

For details on the `last_crossing()` function, see 4.4.10.

## 10.9 Interpolation Function

This section details the table model task.

### 10.9.1 The `$table_model` function

The `$table_model` function models the behavior of a design by interpolating between data points. To build such a model, the user needs to provide a set of sample points  $(x_{i1}, x_{i2}, \dots, x_{iN}, y_i)$  so that  $f(x_{i1}, x_{i2}, \dots, x_{iN}) = y_i$ , where  $f$  is the model function and  $N$  is the number

of independent variables of the model. Using interpolation techniques, one can get the model value at any point in the domain of the sample points approximately. In the case when the evaluated point is outside the domain of the sample points, extrapolation techniques can be used to calculate the model value at that point. However, extrapolation is usually unreliable when the evaluated point is far from the domain of the sample points.

The syntax for this task is shown in Syntax 10-9.

```

table_model_declaration ::= $table_model (variables, data_source, <ctrl_string >)
variables                ::= independent_var {, independent_var}
independent_var         ::= numerical_expression
data_source             ::= data_file | data_array
data_file               ::= "filename"
data_arrays             ::= array1 [, array2 [, array3]], output_array
                        | multi_dimensional_array
ctrl_string             ::= "<sub_ctrl_string {',' sub_ctrl_string}>"
sub_ctrl_string         ::= <degree_char> <extrap_char <extrap_char>>
degree_char             ::= '1' | '2' | '3'

```

Syntax 10-9—Syntax for table model task

### 10.9.2 Table Model Variables

`variables` are numerical expressions that are used as the independent model variables. They can be any legal expressions that can be assigned to an analog signal.

### 10.9.3 Data Source

`data_source` specifies the source of sample points for the table model. There are two kinds of data sources: files and arrays. The file source requires that the sample points be stored in a file, while the array source requires that the data points be stored in a set of array variables. The user can select the data source by either providing the file name of a file source or a set of array variables.

For a file source, conceptually the sample points are stored in the following format:

$$P_1 P_2 P_3 \dots P_M$$

where  $P_i$  ( $i=1\dots M$ ) are the sample points. Each sample point  $P_i$  is represented as a sequence of numbers in the order of  $X_{i1} X_{i2} \dots X_{iN} Y_i$ , where  $X_{ik}$  is the coordinate of the sample point in  $k$ 'th dimension and  $Y_i$  is the model value at this point. To increase readability of the data file, comments can occur in any place of the file. Comments begin with '#'.

The sample points can be stored in the file in any order. However, the lexical ascending order is recommended for performance reason. The lexical order  $P_i < P_j$  holds if and only if there exists  $k$ , so that  $X_{il} = X_{jl}$  ( $l = 1 \dots k-1$ ) and  $X_{ik} < X_{jk}$ . If the sample points are not in the lexical order, the system will automatically sort the data file. The following shows an sample data file:

```
# Id = Is * (exp (Vd / (n * Vt)) - 1.0)
#      Is = 1e-14, Vt = 0.025875, n = 1
#
#      Id          vd
# -----
#      0.000000e+00  0.000000e+00
#      1.000000e-01  4.669060e-13
#      2.000000e-01  2.273393e-11
#      3.800000e-01  2.388055e-08
#      4.000000e-01  5.172864e-08
#      4.800000e-01  1.138878e-06
#      5.600000e-01  2.507400e-05
#      6.400000e-01  5.520390e-04
#      7.200000e-01  1.215391e-02
#      7.400000e-01  2.632707e-02
#      7.600000e-01  5.702813e-02
#      7.800000e-01  1.235309e-01
#      8.000000e-01  2.675853e-01
#      8.200000e-01  5.796273e-01
#      8.400000e-01  1.255554e+00
#      8.600000e-01  2.719706e+00
#      8.800000e-01  5.891264e+00
#      9.000000e-01  1.276130e+01
#      9.200000e-01  2.764277e+01
#      9.400000e-01  5.987812e+01
#      9.600000e-01  1.297044e+02
#      9.800000e-01  2.809579e+02
#      1.000000e+00  6.085942e+02
#      1.020000e+00  1.318300e+03
#      1.040000e+00  2.855623e+03
#      1.060000e+00  6.185680e+03
#      1.080000e+00  1.339905e+04
#      1.100000e+00  2.902422e+04
#      1.120000e+00  6.287053e+04
#      1.140000e+00  1.361864e+05
#      1.160000e+00  2.949988e+05
#      1.180000e+00  6.390088e+05
```

If the user choose the array source, a set of one-dimensional arrays that contains the data points should be passed to the system task. The size of these arrays is the number of sample points in the table,  $M$ . The data will be stored in the arrays such that for the  $k^{\text{th}}$  dimension of the  $i^{\text{th}}$  sample point,  $\text{array\_k}[i] = X_{ik}$  and such that for the  $i^{\text{th}}$  sample point  $\text{output\_array}[i] = Y_i$ .

If multi-dimension arrays are supported by the simulator, a multi-dimensional array may be passed instead of a set of arrays. That array should be two dimensional. The first dimension is designated for points and the second dimension is is designated for coordinates of the point and its values. Suppose  $p$  is such an array,  $p[i][j]$  contains the value of the  $j$ 'th coordinate of  $i$ 'th point when  $0 \leq j < N$  and  $p[i][N]$  contains the value of the sample point, where  $N$  is the dimension of the sample point.

#### 10.9.4 Control String


The control string is used to control the numerical aspects of the interpolation process. It consists of subcontrol strings for each dimension. The subcontrol string may contain one degree character and one or two extrapolation method characters. The degree is a digit representing the degrees of the splines used for the interpolation. The degree should not exceed 3. When the degree character does not appear in the subcontrol string, degree 1 (i.e. linear interpolation) is assumed.

The extrapolation method controls how the point is evaluated when the point is beyond the region of the user provided sample points. Three extrapolation methods are supported (see Figure 10.1). The Clamp extrapolation method uses a horizontal line that passes through the nearest sample point, also called the end point, to extend the model evaluation. The Linear extrapolation method models the extrapolation through a tangent line at the end point. The Spline extrapolation method uses the polynomial for the nearest segment (the segment at the end) to evaluate a point beyond the interpolation area. The user can also disable extrapolation by choosing the Error extrapolation method. In this case, when the system tries to evaluate a point beyond the interpolation region the system will terminate.<sup>1</sup> The extrapolation method characters used to specify the extrapolation methods are shown in Table 2.

extrapolation character	meaning of the character
C	Clamp extrapolation
L	Linear extrapolation (default)
S	Spline extrapolation
E	Error extrapolation

Users can use up to 2 extrapolation method characters to specify the extrapolation method used for each end. When no extrapolation method character is given, the Linear extrapolation method will be used for both ends as default. When one extrapolation method character is given, the specified extrapolation method will be used for both ends. When two extrapolation method characters are given, the first character specifies the

1. Vendors can provide vendor dependent way to issue warning messages when the evaluated point is beyond the interpolation region.



**Figure 10 Using the CLAMP and LINEAR end conditions**

extrapolation method used for the end with smaller coordinate value, and the second character is used for the end with larger coordinate value.

The following are some examples of the control strings for a 2D table model. In the table D1 and D2 are the interpolation degrees for the first and second dimension. EL1 and ER1 are the extrapolation methods used for the left end and the right end for the first dimension. EL2 and ER2 are the extrapolation methods used for the left end and the right end for the second dimension.

control string	D1	EL1	ER1	D2	EL2	ER2
"1CL,3SL"	1	Clamp	Linear	3	Spline	Linear
"1CL,3SL"	1	Clamp	Linear	3	Spline	Linear
"1L,3S"	1	Linear	Linear	3	Spline	Spline
"1,3CL"	1	Linear	Linear	3	Clamp	Linear
"1"	1	Linear	Linear	1	Linear	Linear
","3"	1	Linear	Linear	3	Linear	Linear
""	1	Linear	Linear	1	Linear	Linear

### 10.9.5 An Example

The following example shows a simplified table model for the NMOS.

```
'include "discipline.h"  
'include "constants.h"  
  
module mynmos (g, d, s);  
  electrical g, d, s;  
  inout     g, d, s;  
  analog begin  
    I(d, s) <+ $table_model (V(g, s), V(d, s), "nmos.tbl", "3CL,3CL");  
  end  
endmodule
```

